

STAT 5200 Handout #30: Restricted vs. Unrestricted Models in a Mixed Effects Design (Ch. 12)

Example: (Data as reported in Table 13.13 of Cobb's 1998 text "Introduction to Design and Analysis of Experiments) The goal of a particular study is to assess the effects of various factors on the hardness of gold fillings in teeth. Three factors (and their levels) are as follows:

- **Dentist** : 1-5, chosen from a certain School of Dentistry
- **Method**: 1-3
 - 1: Condensing (special hand tool to pack gold)
 - 2: Hand-malleting (special hand tool to pack gold, tapped with small hammer by assistant)
 - 3: Mechanical-malleting (special hand tool to pack gold, tapped with small hammer built into tool)
- **Type** (of gold): 1-2
 - 1: Gold foil (tiny cylinders of pure gold)
 - 2: Goldent (powdered gold in a gold foil envelope)

At each Dentist/Method/Type combination, ten cavities (previously drilled into a single block of ivory) were filled (by the dentist, using the method, and with the type of gold), and a **Hardness** reading was taken for each one using a pyramid-shaped diamond pressed into the filling. The sum of the sizes of the ten indentations were recorded.

This is a $5 \times 3 \times 2$ factorial experiment with one observation per combination of factor levels. Method is a fixed factor, but Dentist and Type can be considered random factors.

```
/* Enter data */
data gold;
  input Dentist Method Type Hardness @@; cards;
1 1 1 792 1 1 2 824 1 2 1 772 1 2 2 772
1 3 1 782 1 3 2 803
2 1 1 803 2 1 2 803 2 2 1 752 2 2 2 772
2 3 1 715 2 3 2 707
3 1 1 715 3 1 2 724 3 2 1 792 3 2 2 715
3 3 1 762 3 3 2 606
4 1 1 673 4 1 2 946 4 2 1 657 4 2 2 743
4 3 1 690 4 3 2 245
5 1 1 634 5 1 2 715 5 2 1 649 5 2 2 724
5 3 1 724 5 3 2 627
;

/* Obtain EMS and MS */
proc mixed data=gold method=type3;
  class dentist method type;
  model hardness = method / ddfm=kr ;
  random dentist type
          dentist*method dentist*type method*type;
  title1 'Getting EMS and MS';
run;
```

Getting EMS and MS

Type 3 Analysis of Variance

Source	DF	Sum of Squares	Mean Square	Expected Mean Square
Method	2	49598	24799	Var(Residual) + 5 Var(Method*Type) + 2 Var(Dentist*Method) + Q(Method)
Dentist	4	71527	17882	Var(Residual) + 3 Var(Dentist*Type) + 2 Var(Dentist*Method) + 6 Var(Dentist)
Type	1	1153.200000	1153.200000	Var(Residual) + 5 Var(Method*Type) + 3 Var(Dentist*Type) + 15 Var(Type)
Dentist*Method	8	87219	10902	Var(Residual) + 2 Var(Dentist*Method)
Dentist*Type	4	9514.466667	2378.616667	Var(Residual) + 3 Var(Dentist*Type)
Method*Type	2	62453	31227	Var(Residual) + 5 Var(Method*Type)
Residual	8	93789	11724	Var(Residual)

Source	Error Term	Error DF	F Value	Pr > F
Method	MS(Dentist*Method) + MS(Method*Type) - MS(Residual)	1.7793	0.82	0.5605
Dentist	MS(Dentist*Method) + MS(Dentist*Type) - MS(Residual)	0.0725	11.48	0.8201
Type	MS(Dentist*Type) + MS(Method*Type) - MS(Residual)	0.946	0.05	0.8580
Dentist*Method	MS(Residual)	8	0.93	0.5396
Dentist*Type	MS(Residual)	8	0.20	0.9297
Method*Type	MS(Residual)	8	2.66	0.1298
Residual

```

/* Get F-statistics and p-values */
data temp; input Source $
  numDF numMS denDFu denMSu denDFr denMSr; cards;
D 4 17882 0.0725 1556.617 4 2378.617
T 1 1153.2 0.946 21881.617 4 2378.617
;
data temp; set temp;
Fu = numMS/denMSu; pval_u = 1-cdf('F',Fu,numDF,denDFu);
Fr = numMS/denMSr; pval_r = 1-cdf('F',Fr,numDF,denDFr);
proc print data=temp;
var Source numDF Fu denDFu Fr denDFr pval_u pval_r;
title1 'Significance Results for Gold Teeth Data';
run;

```

Significance Results for Gold Teeth Data								
Obs	Source	numDF	Fu	denDFu	Fr	denDFr	pval_u	pval_r
1	D	4	11.4877	0.0725	7.51781	4	0.82006	0.03811
2	T	1	0.0527	0.9460	0.48482	4	0.85797	0.52459

```

/* Get restricted model test of Dentist
by treating Dentist as fixed */
proc mixed data=gold method=type3 plots=(residualpanel);
class dentist method type;
model hardness = method|dentist / ddfm=kr;
random type method*type dentist*type;
title1 'Restricted Model test of Dentist';
title2 '(by treating Dentist as fixed)';
run;

```

Restricted Model test of Dentist (by treating Dentist as fixed)				
Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Method	2	2	0.00	1.0000
Dentist	4	4	7.52	0.0381
Dentist*Method	8	8	0.93	0.5396

(Handout #30, Continued – EMS Calculation, after first constructing Hasse Diagram)

<i>For each subscript, indicate fixed or random, and number of levels</i>						
				Variance		
				Component		
Source	R-5	F-3	R-2	Product	Component	EMS
	i	j	k			
D _i	1	3	2	6	σ_D^2	$6\sigma_D^2 + 2\sigma_{DM}^2 + 3\sigma_{DT}^2 + \sigma^2$
M _j	5	1	2	10	Q _M	$10Q_M + 2\sigma_{DM}^2 + 5\sigma_{MT}^2 + \sigma^2$
DM _{ij}	1	1	2	2	σ_{DM}^2	$2\sigma_{DM}^2 + \sigma^2$
T _k	5	3	1	15	σ_T^2	$15\sigma_T^2 + 3\sigma_{DT}^2 + 5\sigma_{MT}^2 + \sigma^2$
DT _{ik}	1	3	1	3	σ_{DT}^2	$3\sigma_{DT}^2 + \sigma^2$
MT _{jk}	5	1	1	5	σ_{MT}^2	$5\sigma_{MT}^2 + \sigma^2$
E _{ijk}	1	1	1	1	σ^2	σ^2
<i>1 if row contains subscript; number of levels otherwise</i>				<i>across all subscript columns</i>	<i>Q for fixed; σ^2 for random</i>	<i>sum of “contribution” for Source and every [eligible] term below it in the Hasse Diagram</i>
				<i>Product*(Variance Component) is the “contribution” of the term</i>	<i>NOTE: In <u>restricted</u> model, mixed interaction terms containing fixed factors not found in Source are not “eligible”</i>	